

Lane2 Language Specification

Version: v1 draft

Status: working specification

Contents

1	Introduction	2
1.1	Scope	2
1.2	Compatibility	2
1.3	Experimental Features	2
1.4	Feedback	2
1.5	Reference	3
2	Syntax and Grammar	3
2.1	Notation	3
2.2	Lexical Grammar	3
2.3	Syntax Grammar	6
2.4	Comments	9
3	Type System	11
3.1	Notations	11
3.2	Primitive Types	11
3.3	Function Types	13
3.4	Generic Function Types	13
3.5	Nominal Custom Types	14
3.6	Struct Types	15
3.7	Enum Types	15
4	Type Inference	16
5	Builtins	16
5.1	Required Ininsics	17
6	Prelude	17
6.1	Standard Operation Structs	17
6.2	Standard Preopen Values	18
7	Declarations	19
7.1	Top-Level Declarations	19
7.2	Struct Declarations	19
7.3	Enum Declarations	19
7.4	Function Declarations	19
7.5	Let Declarations	20
8	Scopes and Bindings	20
8.1	Namespaces	21
8.2	Local Bindings	21
8.3	Open Scope Extensions	21
8.4	Preopen	22
9	Expressions	22
9.1	Blocks	22
9.2	Conditional Expressions	23
9.3	Calls, Field Access, and Pipeline	23
10	Structs, Enums, and Open	24
10.1	Enums	25
10.2	Openable Struct Values	26
10.3	Struct Field Forwarding	26
11	Pattern Matching	26
12	Operators	27
13	Evaluation	28

14	Undefined Behavior	28
15	Conformance	29
16	Appendix	29
16.1	Deliberately Omitted From V1	29

1 Introduction

Lane2 is a strict, pure, expression-oriented functional programming language. Its first implementation target is a parser, a type checker, and an AST interpreter; later implementations may add bytecode compilation, a virtual machine, a linker, and effect handling.

Lane2 is intentionally small. It has no mutable state, assignment, trait system, method dispatch, module system, or implicit typeclass-style instance search in v1. Instead, v1 is centered on nominal data, first-class functions, bidirectional local type inference, exhaustive pattern matching, and explicit operation values opened into lexical scope.

This document specifies Lane2/Core v1: the platform-independent part of Lane2 that should behave the same across the initial AST interpreter and later execution backends.

1.1 Scope

Lane2/Core v1 covers:

- source structure and declarations;
- lexical scope and name resolution;
- nominal struct and enum types;
- function and block expressions;
- local type inference;
- pattern matching;
- open, preopen, and operation-based operators;
- unsafe builtin expressions.

The following are outside Lane2/Core v1:

- mutation and assignment;
- modules and imports;
- bytecode and virtual machine semantics;
- linker entrypoint selection;
- algebraic effects;
- type aliases;
- traits, typeclasses, and interfaces;
- tuples and collection syntax.

1.2 Compatibility

This specification is a v1 draft. No compatibility is promised between draft revisions. Language rules may be added, removed, or changed while the first implementation is being developed.

Once a v1 implementation is declared stable, this section should be replaced by an explicit compatibility policy.

1.3 Experimental Features

This draft does not distinguish stable and experimental language features. Every rule in this document is provisional until v1 is stabilized.

Future drafts may mark individual features as experimental when their surface syntax or semantics are intentionally unsettled.

1.4 Feedback

Design feedback is tracked in the Lane2 repository. Implementation changes should update this specification, docs/language-v1.md, docs/prelude-v1.md, CONTEXT.md, and ADRs when the change affects user-visible semantics.

1.5 Reference

When referencing this draft, use:

> Lane2 Language Specification : Lane2/Core v1 draft.

2 Syntax and Grammar

2.1 Notation

```
grammar      ::= { production }
production   ::= nonTerminal "::=" grammarExpression
grammarExpression ::=
    grammarSequence { "|" grammarSequence }
grammarSequence ::=
    { grammarItem }
grammarItem ::=
    terminal
    | nonTerminal
    | "[" grammarExpression "]"
    | "{" grammarExpression "}"
    | "(" grammarExpression ")"
    | grammarItem "?"
    | grammarItem "*"
    | grammarItem "+"

terminal     ::= "'" terminalText "'"
nonTerminal  ::= lowerCamelIdentifier
```

empty sequence denotes epsilon.

"A B" denotes sequencing.

"A | B" denotes choice.

"[A]" and "A?" denote optional occurrence.

"{ A }" and "A*" denote zero or more occurrences.

"A+" denotes one or more occurrences.

Parenthesized grammar expressions group without producing syntax.

2.2 Lexical Grammar

```
lexicalInput ::=
    (trivia | token)* eof

trivia ::=
    whitespace
    | lineTerminator
    | comment

token ::=
    keyword
    | reservedWord
    | identifier
    | intLiteral
    | stringLiteral
    | boolLiteral
    | operatorToken
    | punctuationToken

keyword ::=
    "struct"
    | "enum"
    | "fn"
    | "let"
```

```

    | "open"
    | "if"
    | "else"
    | "match"
    | "builtin"

reservedWord ::=
    "effect"
    | "handler"
    | "module"
    | "import"
    | "pub"
    | "type"
    | "trait"
    | "interface"
    | "mut"
    | "return"

identifier ::=
    xidStart identifierContinue*
    | "_" identifierContinue+

identifierContinue ::=
    "_"
    | xidContinue
    | decimalDigit

boolLiteral ::=
    "true"
    | "false"

intLiteral ::=
    decimalDigit+

stringLiteral ::=
    "\"" stringElement* "\""

stringElement ::=
    asciiStringCharacter
    | stringEscape

asciiStringCharacter ::=
    asciiCodePointExceptControlBackslashQuote

stringEscape ::=
    "\\" " \" \"
    | "\\" "\ \" \"
    | "\\" "n"
    | "\\" "r"
    | "\\" "t"
    | "\\" "x" asciiHexByte

asciiHexByte ::=
    asciiHexLowByte
    | asciiHexHighByte

asciiHexLowByte ::=
    ("0" | "1" | "2" | "3" | "4" | "5" | "6") asciiHexDigit

asciiHexHighByte ::=
    "7" ("0" | "1" | "2" | "3" | "4" | "5" | "6" | "7")

```

```

operatorToken ::=
    "+"
  | "-"
  | "*"
  | "/"
  | "%"
  | "=="
  | "!="
  | "<"
  | "<="
  | ">"
  | ">="
  | "&&"
  | "||"
  | "!"
  | ">"

punctuationToken ::=
    "("
  | ")"
  | "{"
  | "}"
  | "["
  | "]"
  | ","
  | "."
  | ":"
  | "::"
  | "->"
  | "=>"
  | "="
  | "-"

whitespace ::=
    U+0009
  | U+000B
  | U+000C
  | U+0020

lineTerminator ::=
    U+000A
  | U+000D
  | U+000D U+000A

decimalDigit ::=
    "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9"

asciiHexDigit ::=
    decimalDigit
  | "a" | "b" | "c" | "d" | "e" | "f"
  | "A" | "B" | "C" | "D" | "E" | "F"

xidStart ::=
    any Unicode code point with the XID_Start property

xidContinue ::=
    any Unicode code point with the XID_Continue property

```

Lexical analysis uses maximal munch. If more than one token class matches the same maximal source span, keyword, reservedWord, and boolLiteral take priority over identifier.

2.3 Syntax Grammar

```
sourceFile ::=
    topLevelDeclaration* eof

topLevelDeclaration ::=
    structDeclaration
  | enumDeclaration
  | functionDeclaration
  | topLevelLetDeclaration
  | anonymousTopLevelLetDeclaration
  | openDeclaration

structDeclaration ::=
    "struct" typeName typeParameters? "{" structMember* "}"

structMember ::=
    fieldDeclaration
  | fieldForwardingDeclaration

fieldDeclaration ::=
    fieldName space ":" space type

fieldForwardingDeclaration ::=
    "open" valueName

enumDeclaration ::=
    "enum" typeName typeParameters? "{" enumVariant* "}"

enumVariant ::=
    variantName enumPayload?

enumPayload ::=
    "(" commaSeparatedTypes? ")"

functionDeclaration ::=
    "fn" typeParameters? functionName "(" commaSeparatedParameters? ")" "->" type block

parameter ::=
    valueName space ":" space type

topLevelLetDeclaration ::=
    "let" valueName space ":" space type "=" expression

anonymousTopLevelLetDeclaration ::=
    "let" space ":" space type "=" expression

localLetDeclaration ::=
    "let" valueName typeAnnotation? "=" expression

typeAnnotation ::=
    space ":" space type

openDeclaration ::=
    "open" valueName

type ::=
    typeConstructor typeArguments?
  | functionType

typeConstructor ::=
    typeName
```

```

typeArguments ::=
    "[" commaSeparatedTypes "]"

typeParameters ::=
    "[" commaSeparatedTypeParameters "]"

functionType ::=
    typeParameters? "(" commaSeparatedTypes? ")" "->" type

expression ::=
    ifExpression
    | matchExpression
    | functionLiteral
    | block
    | pipelineExpression

pipelineExpression ::=
    binaryExpression { "|>" pipelineRhs }

pipelineRhs ::=
    callExpression
    | functionLiteral

binaryExpression ::=
    unaryExpression { binaryOperator unaryExpression }

unaryExpression ::=
    unaryOperator unaryExpression
    | callExpression

callExpression ::=
    fieldExpression { callSuffix }

callSuffix ::=
    "(" commaSeparatedExpressions? ")"

fieldExpression ::=
    primaryExpression { "." fieldName }

primaryExpression ::=
    literal
    | valueName
    | qualifiedVariantExpression
    | structLiteral
    | builtinExpression
    | "(" expression ")"
    | "(" ")"

literal ::=
    intLiteral
    | stringLiteral
    | boolLiteral

qualifiedVariantExpression ::=
    typeName typeArguments? "::" variantName variantArguments?

variantArguments ::=
    "(" commaSeparatedExpressions? ")"

structLiteral ::=

```

```

    typeName typeArguments? ":" "{" commaSeparatedStructLiteralFields "}"

structLiteralField ::=
    fieldName
  | fieldName ":" expression

builtinExpression ::=
    "builtin" "(" stringLiteral ")"

functionLiteral ::=
    "fn" typeParameters? "(" commaSeparatedFunctionLiteralParameters? ")"
functionReturnAnnotation? block

functionLiteralParameter ::=
    valueName
  | valueName space ":" space type

functionReturnAnnotation ::=
    "->" type

block ::=
    "{" localItem* expression "}"

localItem ::=
    localLetDeclaration
  | functionDeclaration
  | openDeclaration

ifExpression ::=
    "if" expression block "else" block

matchExpression ::=
    "match" expression "{" matchArm* "}"

matchArm ::=
    pattern "=>" expression

pattern ::=
    "_"
  | valueName
  | literal
  | qualifiedVariantPattern
  | structPattern

qualifiedVariantPattern ::=
    typeName ":" variantName patternArguments?

patternArguments ::=
    "(" commaSeparatedPatterns? ")"

structPattern ::=
    typeName ":" "{" commaSeparatedStructPatternFields "}"

structPatternField ::=
    fieldName
  | fieldName ":" pattern

binaryOperator ::=
    "+" | "-" | "*" | "/" | "%"
  | "==" | "!=" | "<" | "<=" | ">" | ">="
  | "&&" | "||"

```



```

unaryOperator ::=
    "-" | "!"

commaSeparatedTypes ::=
    type ("," type)* ","?

commaSeparatedTypeParameters ::=
    typeParameter ("," typeParameter)* ","?

commaSeparatedParameters ::=
    parameter ("," parameter)* ","?

commaSeparatedExpressions ::=
    expression ("," expression)* ","?

commaSeparatedFunctionLiteralParameters ::=
    functionLiteralParameter ("," functionLiteralParameter)* ","?

commaSeparatedStructLiteralFields ::=
    structLiteralField ("," structLiteralField)* ","?

commaSeparatedPatterns ::=
    pattern ("," pattern)* ","?

commaSeparatedStructPatternFields ::=
    structPatternField ("," structPatternField)* ","?

typeName ::=
    identifier

functionName ::=
    identifier

valueName ::=
    identifier

fieldName ::=
    identifier

variantName ::=
    identifier

typeParameter ::=
    identifier

space ::=
    whitespace+

```

The precedence and associativity of `binaryOperator`, `unaryOperator`, calls, field access, and pipeline expressions are defined in “Operators”.

2.4 Comments

```

comment ::=
    lineComment
    | blockComment

lineComment ::=
    "/" "/" lineCommentCharacter* lineTerminator?

```

```
lineCommentCharacter ::=  
    any Unicode code point except U+000A or U+000D  
  
blockComment ::=  
    "/" "*" blockCommentCharacter* "*" "/"  
  
blockCommentCharacter ::=  
    any Unicode code point sequence that does not start "*/"
```

Comments are trivia. Comments do not appear in the syntactic grammar.

3 Type System

3.1 Notations

Notation	Meaning
T, U, R, F, P, Q	types
A	type variable
C	type constructor
S	struct type constructor
E	enum type constructor
e, c, f, b	expressions
a	match arm
x	value binder
i, j, k, n, m, q	indices and natural numbers
l	struct field name
v	enum variant name
Δ	type-constructor context
Θ	type-variable context
Γ	value typing context
D	custom type definition
$\Delta(C) = D$	visible custom type definition
$\text{struct}(A_1, \dots, A_n; l_1 : F_1, \dots, l_m : F_m)$	struct definition
$\text{enum}\left(A_1, \dots, A_n; v_j(P_{j,1}, \dots, P_{j,m_j})\right)$	enum definition
$\text{params}(D) = (A_1, \dots, A_n)$	custom type parameters
σ	type substitution environment
$\sigma = [T_1/A_1, \dots, T_n/A_n]$	substitution environment binding
$U[T_1/A_1, \dots, T_n/A_n]$	direct type substitution
$U\sigma$	type substitution by environment
$\Delta\Theta \vdash T \text{ type}$	well-formed type
$C[T_1, \dots, T_n]$	nominal type application
$\forall A_1, \dots, A_n. T$	universal type
$T \equiv U$	type equality
$\Gamma \vdash e : T$	expression typing
$\text{fields}(S[T_1, \dots, T_n]) = (l_1 : Q_1, \dots, l_m : Q_m)$	instantiated struct fields
$\text{payloads}(E[T_1, \dots, T_n], v) = (Q_1, \dots, Q_m)$	instantiated variant payloads
$\text{binders}(\Gamma x_1 : Q_1, \dots, x_m : Q_m) = \Gamma'$	extended pattern-binder context
$\text{arm-type}(E[T_1, \dots, T_n], R)$	typed enum match arm
$\frac{P}{Q}$	rule with premise P and conclusion Q
name	abstract syntax constructor

Table 1: Type system notations

3.2 Primitive Types

Lane2/Core v1 has four primitive types: `Unit`, `Bool`, `Int`, and `String`.

Primitive formation. Each primitive type is well-formed in every type environment.

$$\Delta\Theta \vdash \text{Unit} \text{ type}$$
$$\Delta\Theta \vdash \text{Bool} \text{ type}$$
$$\Delta\Theta \vdash \text{Int} \text{ type}$$
$$\Delta\Theta \vdash \text{String} \text{ type}$$

Unit introduction. The expression `()` introduces a `Unit` value.

```
() // unit literal
```

$$\Gamma \vdash () : \text{Unit}$$

Unit elimination. Lane2/Core v1 has no dedicated elimination form for `Unit`.

Bool introduction. Boolean literals introduce `Bool` values.

```
true // boolean literal  
false // boolean literal
```

$$\Gamma \vdash \text{true} : \text{Bool}$$
$$\Gamma \vdash \text{false} : \text{Bool}$$

Bool elimination. `if` eliminates a `Bool` value.

```
if c {  
  e1  
} else {  
  e2  
} // c : Bool, e1 : T, e2 : T
```

$$\frac{\Gamma \vdash c : \text{Bool} \quad \Gamma \vdash e_1 : T \quad \Gamma \vdash e_2 : T}{\Gamma \vdash \text{if}(c, e_1, e_2) : T}$$

Int introduction. Integer literals introduce `Int` values.

```
n // integer literal
```

$$\Gamma \vdash n : \text{Int}$$

Int elimination. Lane2/Core v1 has no built-in syntactic eliminator for `Int`. Integer operations are typed through required intrinsics and prelude operation values.

String introduction. ASCII string literals introduce `String` values.

```
"..." // ASCII string literal
```

$$\Gamma \vdash s : \text{String}$$

String elimination. Lane2/Core v1 has no built-in syntactic eliminator for `String`. String equality is typed through the prelude.

Primitive type equality. Primitive types are equal only to themselves.

$$\text{Unit} \equiv \text{Unit}$$
$$\text{Bool} \equiv \text{Bool}$$
$$\text{Int} \equiv \text{Int}$$
$$\text{String} \equiv \text{String}$$

3.3 Function Types

Function formation.

```
fn f(x1 : T1, ..., xn : Tn) -> R { e } // function definition
let f : (T1, ..., Tn) -> R = fn(x1, ..., xn) { e } // function literal binding
```

$$\frac{\Delta\Theta \vdash T_1 \text{ type} \quad \dots \quad \Delta\Theta \vdash T_n \text{ type} \quad \Delta\Theta \vdash R \text{ type}}{\Delta\Theta \vdash (T_1, \dots, T_n) \rightarrow R \text{ type}}$$

Function introduction.

```
fn(x1 : T1, ..., xn : Tn) -> R { e } // function literal
fn f(x1 : T1, ..., xn : Tn) -> R { e } // function definition
```

$$\frac{\Gamma, x_1 : T_1, \dots, x_n : T_n \vdash e : R}{\Gamma \vdash \text{fn}((x_1 : T_1), \dots, (x_n : T_n), R, e) : (T_1, \dots, T_n) \rightarrow R}$$

Function elimination.

```
f(a1, ..., an) // function call
```

$$\frac{\Gamma \vdash f : (T_1, \dots, T_n) \rightarrow R \quad \Gamma \vdash a_1 : T_1 \quad \dots \quad \Gamma \vdash a_n : T_n}{\Gamma \vdash f(a_1, \dots, a_n) : R}$$

Function type equality.

$$\frac{T_1 \equiv U_1 \quad \dots \quad T_n \equiv U_n \quad R \equiv S}{(T_1, \dots, T_n) \rightarrow R \equiv (U_1, \dots, U_n) \rightarrow S}$$

Function types are uncurried. $(T_1, T_2) \rightarrow R$ is not the same type object as $(T_1) \rightarrow (T_2) \rightarrow R$.

3.4 Generic Function Types

Generic function formation.

```
[A1, ..., An](T1, ..., Tm) -> R // generic function type
```

$$\frac{\Delta\Theta, A_1, \dots, A_n \vdash (T_1, \dots, T_m) \rightarrow R \text{ type}}{\Delta\Theta \vdash \forall A_1, \dots, A_n. (T_1, \dots, T_m) \rightarrow R \text{ type}}$$

Generic function introduction.

```
fn[A1, ..., An](x1 : T1, ..., xm : Tm) -> R { e } // generic function literal
fn[A1, ..., An] f(x1 : T1, ..., xm : Tm) -> R { e } // generic function definition
```

A generic function literal or named generic function introduces a generic function value.

Generic function elimination.

Generic function calls instantiate type parameters at the use site when the use site is unambiguous.

```
f(a1, ..., am) // generic function call with inferred type arguments
```

Generic function type equality.

$$\frac{(T_1, \dots, T_m) \rightarrow R \equiv (U_1, \dots, U_m) \rightarrow S}{\forall A_1, \dots, A_n. (T_1, \dots, T_m) \rightarrow R \equiv \forall A_1, \dots, A_n. (U_1, \dots, U_m) \rightarrow S}$$

Generic function type equality compares the number of type parameters and the function types under corresponding bound type parameters.

3.5 Nominal Custom Types

Struct and enum declarations introduce nominal custom type definitions in Δ .

```
struct S[A1, ..., An] {
  l1 : F1
  ...
  lm : Fm
} // struct definition
```

```
enum E[A1, ..., An] {
  v1(P11, ..., P1m1)
  ...
  vj(Pj1, ..., Pjmj)
} // enum definition
```

The type-constructor context stores the full declaration shape, not only arity.

$$\Delta(S) = \text{struct}(A_1, \dots, A_n; l_1 : F_1, \dots, l_m : F_m)$$

$$\Delta(E) = \text{enum}\left(A_1, \dots, A_n; v_j(P_{j,1}, \dots, P_{j,m_j})\right)$$

Top-level custom type declarations are checked in two phases. First, all top-level struct and enum constructors are collected into Δ with their type parameters, field names, variant names, and declared member type expressions. Second, every field type and variant payload type is checked under the collected Δ and the declaration's type parameters. This permits mutually recursive custom types.

$$\frac{\Delta(S) = \text{struct}(A_1, \dots, A_n; l_1 : F_1, \dots, l_m : F_m) \quad \forall i. \Delta A_1, \dots, A_n \vdash F_i \text{ type}}{\Delta \vdash S \text{ declaration}}$$

$$\frac{\Delta(E) = \text{enum}\left(A_1, \dots, A_n; v_j(P_{j,1}, \dots, P_{j,m_j})\right) \quad \forall j, k. \Delta A_1, \dots, A_n \vdash P_{j,k} \text{ type}}{\Delta \vdash E \text{ declaration}}$$

Nominal formation. A nominal type application is well-formed when its custom type constructor is visible, the number of type arguments matches the declaration’s type parameters, and every type argument is well-formed.

$$\frac{\Delta(C) = D \quad \text{params}(D) = (A_1, \dots, A_n) \quad \Delta\Theta \vdash T_1 \text{ type} \quad \dots \quad \Delta\Theta \vdash T_n \text{ type}}{\Delta\Theta \vdash C[T_1, \dots, T_n] \text{ type}}$$

Nominal type equality. Nominal type equality compares constructor identity and then compares type arguments pairwise. There is no equality rule whose conclusion relates different nominal constructors.

$$\frac{T_1 \equiv U_1 \quad \dots \quad T_n \equiv U_n}{C[T_1, \dots, T_n] \equiv C[U_1, \dots, U_n]}$$

3.6 Struct Types

Struct introduction. A qualified struct literal introduces a struct value. It must provide every declared field exactly once. Source field order is not significant; the rule below uses declaration order.

```
S[T1, ..., Tn]::{ l1: e1, ..., lm: em } // qualified struct literal
```

$$\frac{\Delta(S) = \text{struct}(A_1, \dots, A_n; l_1 : F_1, \dots, l_m : F_m) \quad \sigma = [T_1/A_1, \dots, T_n/A_n]}{\text{fields}(S[T_1, \dots, T_n]) = (l_1 : F_1\sigma, \dots, l_m : F_m\sigma)}$$

$$\frac{\text{fields}(S[T_1, \dots, T_n]) = (l_1 : Q_1, \dots, l_m : Q_m) \quad \Gamma \vdash e_1 : Q_1 \quad \dots \quad \Gamma \vdash e_m : Q_m}{\Gamma \vdash \text{struct-lit}(S[T_1, \dots, T_n], l_1 = e_1, \dots, l_m = e_m) : S[T_1, \dots, T_n]}$$

Struct field elimination. Field access eliminates a struct value by selecting a declared field type after substituting the struct type arguments.

```
e.l // field access
```

$$\frac{\Gamma \vdash e : S[T_1, \dots, T_n] \quad \text{fields}(S[T_1, \dots, T_n]) = (\dots, l : Q, \dots)}{\Gamma \vdash e.l : Q}$$

Struct values are also eliminated by struct patterns and exposed by open; their detailed static rules are specified in “Pattern Matching” and “Structs, Enums, and Open”.

3.7 Enum Types

Enum variant introduction. A qualified enum variant expression introduces an enum value. The payload expressions must match the declared variant payload types after substituting the enum type arguments.

```

E[T1, ..., Tn]::v(e1, ..., em) // qualified variant expression
v(e1, ..., em) // unqualified variant expression after unambiguous resolution

```

$$\frac{\Delta(E) = \text{enum}(A_1, \dots, A_n; \dots, v(P_1, \dots, P_m), \dots)}{\text{payloads}(E[T_1, \dots, T_n], v) = (P_1[T_1/A_1, \dots, T_n/A_n], \dots, P_m[T_1/A_1, \dots, T_n/A_n])}$$

$$\frac{\text{payloads}(E[T_1, \dots, T_n], v) = (Q_1, \dots, Q_m) \quad \Gamma \vdash e_1 : Q_1 \quad \dots \quad \Gamma \vdash e_m : Q_m}{\Gamma \vdash \text{variant}(E[T_1, \dots, T_n], v, e_1, \dots, e_m) : E[T_1, \dots, T_n]}$$

An unqualified enum variant expression is typed by the same rule after name resolution has selected exactly one visible enum variant.

Enum match elimination. A match expression eliminates an enum value. For an enum scrutinee type, every declared variant must be covered.

```

match e {
  E::v1(x1l1, ..., x1ml) => b1
  ...
  E::vq(xq1, ..., xqmq) => bq
}

```

$$\frac{\text{payloads}(E[T_1, \dots, T_n], v) = (Q_1, \dots, Q_m)}{\text{binders}(\Gamma x_1 : Q_1, \dots, x_m : Q_m) = \Gamma'}$$

$$\frac{\text{payloads}(E[T_1, \dots, T_n], v) = (Q_1, \dots, Q_m) \quad \text{binders}(\Gamma x_1 : Q_1, \dots, x_m : Q_m) = \Gamma' \quad \Gamma' \vdash b : R}{\Gamma \vdash \text{arm}(v(x_1, \dots, x_m) \Rightarrow b) : \text{arm-type}(E[T_1, \dots, T_n], R)}$$

$$\frac{\Gamma \vdash e : E[T_1, \dots, T_n] \quad \Delta(E) = \text{enum}(A_1, \dots, A_n; v_1, \dots, v_q) \quad \Gamma \vdash a_j : \text{arm-type}(E[T_1, \dots, T_n], R)}{\Gamma \vdash \text{match}(e; a_1, \dots, a_q) : R}$$

Pattern syntax and exhaustiveness diagnostics are specified in “Pattern Matching”.

4 Type Inference

Lane2 permits local type inference only at syntactic positions where the omitted type is determined locally.

A binding’s type is not determined from later uses of the bound name.

Local `let` bindings may omit type annotations only when the initializer has a type without using later references to the bound name.

Function literals may omit parameter types only when the immediately surrounding context provides a function type. Otherwise, every value parameter of a function literal must have an explicit type annotation.

Generic function literals without an immediately surrounding generic function type must explicitly declare type parameters and explicitly annotate value parameters.

Generic function calls and generic data constructors may instantiate type parameters at the use site when the use site is unambiguous.

5 Builtins

`builtin("...")` is an unsafe expression.

The checker does not interpret intrinsic strings. Instead, `builtin` receives its type from direct expected context:

```
fn int_add(a : Int, b : Int) -> Int {  
  builtin("%i64_add")  
}
```

This is valid because the function return type provides the expected type `Int` for the body expression.

This is invalid:

```
let x = builtin("%anything")
```

There is no direct expected type.

Incorrect builtin use can produce undefined behavior. Lane2's type safety guarantee applies only to programs that do not misuse builtin.

5.1 Required Ininsics

A conforming Lane2/Core v1 implementation provides these portable intrinsic names:

Intrinsic	Expected type
<code>%i64_add</code>	<code>(Int, Int) -> Int</code>
<code>%i64_sub</code>	<code>(Int, Int) -> Int</code>
<code>%i64_mul</code>	<code>(Int, Int) -> Int</code>
<code>%i64_div</code>	<code>(Int, Int) -> Int</code>
<code>%i64_rem</code>	<code>(Int, Int) -> Int</code>
<code>%i64_neg</code>	<code>(Int) -> Int</code>
<code>%i64_equal</code>	<code>(Int, Int) -> Bool</code>
<code>%i64_less</code>	<code>(Int, Int) -> Bool</code>
<code>%string_equal</code>	<code>(String, String) -> Bool</code>

Table 2: Required intrinsic names and expected types

Other intrinsic names are implementation-defined unsafe builtins.

`%bool_and`, `%bool_or`, `%bool_not`, and `%bool_equal` are not required intrinsics. The standard prelude defines boolean operations as ordinary Lane2 functions and anonymous operation values using `if`; `&&` and `||` supply their right operands as thunks.

6 Prelude

The standard prelude is implementation-supplied Lane2 source checked before user code. It is not a module system.

Prelude declarations provide standard operation structs, primitive wrappers around required intrinsics, derived primitive operations written in Lane2, and anonymous top-level values that populate the initial preopen namespace.

Prelude entries are ordinary Lane2 values and types except where the compiler recognizes operation field pairs for operator aliases.

6.1 Standard Operation Structs

Arithmetic operations:

```
struct Add[T] {  
  add : (T, T) -> T  
}
```

```

struct Sub[T] {
  sub : (T, T) -> T
}

struct Mul[T] {
  mul : (T, T) -> T
}

struct Div[T] {
  div : (T, T) -> T
}

struct Rem[T] {
  rem : (T, T) -> T
}

struct Neg[T] {
  neg : (T) -> T
}

```

Boolean operations:

```

struct And {
  and : (Bool, () -> Bool) -> Bool
}

struct Or {
  or : (Bool, () -> Bool) -> Bool
}

struct Not {
  not : (Bool) -> Bool
}

```

Equality and ordering operations:

```

struct Equal[T] {
  equal : (T, T) -> Bool
  not_equal : (T, T) -> Bool
}

struct Compare[T] {
  equal_impl : Equal[T]
  open equal_impl
  less : (T, T) -> Bool
  less_eq : (T, T) -> Bool
  greater : (T, T) -> Bool
  greater_eq : (T, T) -> Bool
}

```

Operation laws are API conventions, not compiler-checked rules.

6.2 Standard Preopen Values

The prelude may populate the initial preopen namespace with anonymous top-level values. For example, primitive arithmetic and boolean operations can be exposed as follows:

```

let : Add[Int] = Add::{ add: int_add }
let : Sub[Int] = Sub::{ sub: int_sub }
let : Mul[Int] = Mul::{ mul: int_mul }

```

```

let : Div[Int] = Div::{ div: int_div }
let : Rem[Int] = Rem::{ rem: int_rem }
let : Neg[Int] = Neg::{ neg: int_neg }

let int_equal_ops : Equal[Int] = make_equal(int_equal)
let : Compare[Int] = make_compare(int_equal_ops, int_less)

let : And = And::{ and: bool_and }
let : Or = Or::{ or: bool_or }
let : Not = Not::{ not: bool_not }
let : Equal[Bool] = Equal::{
  equal: fn(a : Bool, b : Bool) {
    if a {
      b
    } else {
      bool_not(b)
    }
  },
  not_equal: fn(a : Bool, b : Bool) {
    if a {
      bool_not(b)
    } else {
      b
    }
  },
}

let : Equal[String] = make_equal(string_equal)

```

Boolean conjunction, disjunction, negation, and equality do not require boolean intrinsics; they can be defined with ordinary if expressions in the prelude.

7 Declarations

Declarations introduce types, functions, values, and open scope extensions.

7.1 Top-Level Declarations

Top-level declarations are described by the `topLevelDefinition` grammar in “Syntax and Grammar”.

Top-level forms include struct declarations, enum declarations, named function declarations, typed value declarations, anonymous typed value declarations, and open declarations.

7.2 Struct Declarations

Struct declarations use the grammar in “Structs, Enums, and Open”.

7.3 Enum Declarations

Enum declarations use the grammar in “Structs, Enums, and Open”.

7.4 Function Declarations

Functions are uncurried. There is no automatic currying or partial application.

Function definitions use block bodies:

```

functionDeclaration:
  'fn' [ typeParameters ] functionName '(' [ parameter { ',' parameter } [ ',' ] ] ')' '-'
  >' type block

parameter:
  valueName ':' type

```

```
functionLiteral:
  'fn' [ typeParameters ] '(' [ functionLiteralParameter { ',' functionLiteralParameter }
[ ',' ] ] ')' [ '->' type ] block

functionLiteralParameter:
  valueName
| valueName ':' type
```

```
fn add(a : Int, b : Int) -> Int {
  a + b
}
```

There is no `= expr` function body form.

Function result types use `->`, not `:`.

All named functions must state every parameter type and the result type.

Generic named functions place type parameters after `fn` and before the name:

```
fn[A] id(value : A) -> A {
  value
}
```

Function parameters are positional in v1. Labeled function parameters are not supported.

Function literals produce function values:

```
let f : (Int, Int) -> Int = fn(a, b) {
  a + b
}
```

Generic function literals place type parameters after `fn`:

```
let id = fn[A](value : A) {
  value
}
```

7.5 Let Declarations

Named top-level `let` declarations must include type annotations. Anonymous top-level `let` declarations must include type annotations and must have a struct type.

Local `let` declarations may omit type annotations when their initializer can synthesize a type by local type inference.

8 Scopes and Bindings

Top-level type and function definitions form recursive definition groups.

Top-level `struct`, `enum`, and `fn` definitions may refer to each other regardless of textual order.

Top-level value definitions and top-level `open` declarations follow ordered value scope:

- a top-level `let` initializer may refer only to values already available at that declaration point;
- a top-level `open` may open only a value already available at that declaration point;
- a top-level function body may refer to any top-level value or function, even one declared later.

Example:

```
fn read_x() -> Int {  
  x  
}  
  
let x : Int = 1
```

This is valid because the function body sees the complete top-level environment.

```
let y : Int = x  
let x : Int = 1
```

This is invalid because top-level value initializers follow ordered value scope.

8.1 Namespaces

Lane2 has separate type and value namespaces.

Type and value names may use the same spelling:

```
enum Option[A] {  
  none  
  some(A)  
}  
  
let Option : Int = 42
```

Syntax of the form `Type::member` resolves `Type` in the type namespace.

```
let x : Option[Int] = Option::some(1)
```

The `Option` on the left of `::` denotes the enum type, not the value named `Option`.

8.2 Local Bindings

Local `let` bindings are sequential. A local binding is visible only to later items and the final expression in the same block.

Local `let` bindings may omit type annotations when their initializer can synthesize a type by local type inference:

```
let x = 1
```

Local named functions are also sequential. They may call themselves, but local mutually recursive groups and forward references are not supported:

```
fn f(n : Int) -> Int {  
  fn loop(x : Int) -> Int {  
    if x == 0 {  
      0  
    } else {  
      loop(x - 1)  
    }  
  }  
  loop(n)  
}
```

Local value names may shadow earlier local value names.

8.3 Open Scope Extensions

`open value` opens a struct value.

The operand of `open` must be a visible value name with a struct type. It cannot be an arbitrary expression:

```
open ops
```

This is invalid:

```
open make_ops(10)
```

An `open` scope extension exposes the struct field values as unqualified names from the declaration point to the end of the current lexical scope.

At top level, `open` extends to the end of the file:

```
open int_add_ops

fn add_one(x : Int) -> Int {
  x + 1
}
```

Inside blocks, `open` is a local item and must appear before the final expression:

```
{
  open int_add_ops
  x + y
}
```

Local resolution uses the nearest preceding local binding or `open` scope extension, then falls back to `preopen`.

8.4 Preopen

The `preopen` namespace is a default-open namespace populated by anonymous top-level values.

```
let : Add[Int] = Add:: { add: int_add }
```

An anonymous top-level value must have a struct type. It exposes field values, not generated accessors.

Prelude-provided anonymous values are checked before user code, so their `preopen` entries are visible throughout user code.

User-defined anonymous top-level values extend `preopen` from their declaration point to the end of the top-level scope.

`Preopen` conflicts are errors.

Top-level `open` belongs to the global layer and conflicts with ordinary top-level names or `preopened` names. Local `open` may shadow `preopen` inside its lexical scope.

9 Expressions

Expressions compute values. Lane2 v1 is expression-oriented and has no expression statements.

9.1 Blocks

A block expression contains local items followed by one final expression.

```
block:
  '{' { localItem } expression '}'

localItem:
  localLet
  | localFunctionDeclaration
```

```

    | openDeclaration

localLet:
  'let' valueName [ ':' type ] '=' expression

topLevelLet:
  'let' valueName ':' type '=' expression

anonymousTopLevelLet:
  'let' ':' type '=' expression

openDeclaration:
  'open' valueName

```

```

{
  let x = 1
  fn double(n : Int) -> Int {
    n + n
  }
  double(x)
}

```

Allowed local items:

- let,
- named fn,
- open.

Local type definitions are not supported in v1.

A block does not contain expression statements. This is invalid:

```

{
  f(x)
  g(y)
}

```

An empty block is invalid. Write () explicitly when a Unit value is required.

9.2 Conditional Expressions

if is an expression:

```

if condition {
  then_value
} else {
  else_value
}

```

The else branch is mandatory.

Both branches must have the same type.

9.3 Calls, Field Access, and Pipeline

Function call:

```
f(x, y)
```

Field access:

```
ops.add
```

Field access followed by call:

```
ops.add(x, y)
```

This is not method call syntax. Lane2 v1 has no method receiver lookup.

Pipeline syntax is supported:

```
value |> f(a, b)
```

It rewrites to:

```
f(value, a, b)
```

The right-hand side of `|>` must be a call or function literal.

The following are invalid:

```
value |> f
value |> f(_, y)
```

Placeholders are not supported.

10 Structs, Enums, and Open

```
structDeclaration:
  'struct' typeName [ typeParameters ] '{' { structMember } '}'

structMember:
  fieldDeclaration
| fieldForwarding

fieldDeclaration:
  fieldName ':' type

fieldForwarding:
  'open' fieldName

structLiteral:
  typeName [ typeArguments ] '::' '{' structLiteralField { ',' structLiteralField }
  [ ',' ] '}'

structLiteralField:
  fieldName
| fieldName ':' expression
```

Struct literals are qualified:

```
let p : Point = Point::{ x: 1, y: 2 }
```

Struct field punning is allowed:

```
let p : Point = Point::{ x, y }
```

This is equivalent to:


```
let p : Point = Point::{ x: x, y: y }
```

Struct literals must provide every field exactly once.

The following are not supported:

- anonymous record literals,
- default fields,
- spread,
- copy update,
- field update.

Fields are accessed with dot syntax:

```
p.x
```

V1 has no field visibility modifiers. Every struct field is accessible wherever the struct value is accessible.

10.1 Enums

```
enumDeclaration:
  'enum' typeName [ typeParameters ] '{' { enumVariant } '}'

enumVariant:
  variantName [ '(' [ type { ',' type } [ ',' ] ] ')' ]

qualifiedVariantExpression:
  typeName [ typeArguments ] '::' variantName [ '(' [ expression { ',' expression }
[ ',' ] ] ')' ]

unqualifiedVariantExpression:
  variantName [ '(' [ expression { ',' expression } [ ',' ] ] ')' ]
```

Enum variants are scoped to their enum.

Variant names may be lowercase or uppercase. Capitalization has no semantic role.

Payloadless variants are values and are written without call parentheses:

```
let x : Option[Int] = Option::none
```

This is invalid:

```
Option::none()
```

Variant payloads are positional:

```
enum Expr {
  int(Int)
  add(Expr, Expr)
}
```

Labeled variant payloads are not part of v1. Named product data must be represented with a struct:

```
struct Node[A] {
  left : Tree[A]
  value : A
  right : Tree[A]
}
```

```
enum Tree[A] {
  leaf(A)
  node(Node[A])
}
```

In expressions, unqualified variants are allowed when unambiguous:

```
let x : Option[Int] = some(1)
```

If more than one visible enum has a variant named `some`, the unqualified expression is invalid unless another directly local rule disambiguates it. A qualified variant is always allowed:

```
let x : Option[Int] = Option::some(1)
```

In patterns, variants must always be qualified.

10.2 Openable Struct Values

Only struct values can be opened.

Opening a struct value exposes its field values as unqualified bindings according to the scope rules in “Scopes and Bindings”.

Enum values, function values, primitive values, and arbitrary expressions are not openable.

10.3 Struct Field Forwarding

A struct declaration may contain open field entries:

```
struct Compare[T] {
  equal_impl : Equal[T]
  open equal_impl
  less : (T, T) -> Bool
  less_eq : (T, T) -> Bool
  greater : (T, T) -> Bool
  greater_eq : (T, T) -> Bool
}
```

When a value of this struct type is opened, forwarded fields are exposed too.

Struct field forwarding affects only what is exposed by opening the containing struct value. It does not open the field inside ordinary function bodies.

11 Pattern Matching

`match` is an expression:

```
matchExpression:
  'match' expression '{' { matchArm } '}'

matchArm:
  pattern '=>' expression

pattern:
  '_'
  | valueName
  | literal
  | qualifiedVariantPattern
  | structPattern
```

```
qualifiedVariantPattern:
  typeName '::' variantName [ '(' [ pattern { ',' pattern } [ ',' ] ] ')' ]

structPattern:
  typeName '::' '{' structPatternField { ',' structPatternField } [ ',' ] '}'

structPatternField:
  fieldName
  | fieldName ':' pattern
```

```
match value {
  Option::none => fallback
  Option::some(x) => x
}
```

Match arms use pattern => expression.

Every match must be exhaustive.

V1 patterns:

- wildcard pattern: `_`,
- variable binding pattern: `name`,
- literal pattern,
- qualified enum variant pattern,
- qualified struct pattern.

Enum variants in patterns must be qualified:

```
Option::some(x)
Option::none
```

Bare identifiers in patterns are always variable bindings:

```
match value {
  x => x
}
```

Struct patterns use qualified syntax and must list all fields:

```
match p {
  Point::{ x, y } => x + y
}
```

Struct pattern punning is allowed. Explicit renaming is allowed:

```
match p {
  Point::{ x: a, y: b } => a + b
}
```

Rest patterns, spread patterns, guards, or-patterns, as-patterns, and is pattern expressions are not supported in v1.

12 Operators

Operators are aliases for recognized prelude operation fields.

There is no type-directed instance search. An operator is available only when the relevant operation field is available through preopen or an open scope extension.

Primitive operators are not special-cased. For example, `1 + 2` requires an available `Add::add` operation for `Int`.

Recognized operator mappings:

Operator	Operation
<code>+</code>	<code>Add::add</code>
<code>-</code>	<code>Sub::sub</code>
<code>*</code>	<code>Mul::mul</code>
<code>/</code>	<code>Div::div</code>
<code>%</code>	<code>Rem::rem</code>
unary <code>-</code>	<code>Neg::neg</code>
<code>==</code>	<code>Equal::equal</code>
<code>!=</code>	<code>Equal::not_equal</code>
<code><</code>	<code>Compare::less</code>
<code><=</code>	<code>Compare::less_eq</code>
<code>></code>	<code>Compare::greater</code>
<code>>=</code>	<code>Compare::greater_eq</code>
<code>&&</code>	<code>And::and</code> with a thunked right operand
<code> </code>	<code>Or::or</code> with a thunked right operand
<code>!</code>	<code>Not::not</code>

Table 3: Recognized operator mappings

`&&` and `||` are short-circuit boolean operators. They are recognized mappings to `And::and` and `Or::or`, but the right operand is passed as a zero-argument function instead of being evaluated before the operation call.

`a && b` desugars to a call equivalent to `and(a, fn() { b })` after resolving an available `And::and` operation. `a || b` follows the same rule with `Or::or`. Both operators are defined only for `Bool`.

Concrete expression precedence, associativity, and unary/binary disambiguation follow the MoonBit parser reference where Lane2 has not deliberately removed a feature.

13 Evaluation

Lane2 v1 is pure and strict.

- Evaluating a safe expression depends only on its inputs and produces a value.
- Function arguments are evaluated before the function body runs.
- `if` and `match` evaluate only the selected branch.
- There is no mutable binding, mutable field, assignment, field update, or implicit statement sequencing.
- IO and other effects are not part of v1 core semantics.

The `Unit` value is written explicitly as `()`.

Empty blocks are invalid. A block must contain exactly one final expression after any local items.

Function calls evaluate all arguments before entering the function body. The only v1 operator forms that delay a subexpression are `&&` and `||`, which pass the right operand as a thunk to the resolved prelude operation.

14 Undefined Behavior

Incorrect builtin use can produce undefined behavior. Lane2's type safety guarantee applies only to programs that do not misuse builtin.

Invalid integer arithmetic is undefined behavior in v1. This includes signed 64-bit overflow, division by zero, remainder by zero, `MIN_INT / -1`, and negating `MIN_INT`.

15 Conformance

The words “must”, “must not”, “may”, and “should” are normative in this document.

Text marked as rationale, examples, or notes is informative unless it explicitly uses normative language.

An implementation conforms to this draft if it accepts all valid programs described here, rejects invalid programs described here, and gives the specified typing and evaluation behavior for safe programs.

Programs that misuse `builtin` are outside Lane2’s safety guarantee.

16 Appendix

16.1 Deliberately Omitted From V1

V1 omits:

- mutation and assignment,
- field update and copy update,
- modules and imports,
- local type definitions,
- labeled function parameters,
- labeled enum payloads,
- type aliases,
- traits, typeclasses, interfaces,
- method syntax,
- tuple types,
- collections,
- pattern guards,
- or-patterns,
- as-patterns,
- `is` pattern expressions,
- placeholder pipeline.